KOMTEK I
MICROCOMPUTER



KOMTEK I
MICROCOMPUTER

CONTROL FUNCTION MANUAL

KOMTEK TECHNOLOGIES LTD.

BRIEF SUMMARY OF CONTROL FUNCTION

OPERATION USING DEMONSTRATION TAPE PROGRAMME

1. Load Control Function demonstration tape using CLOAD, press RETURN key.
2. After loading type RUN & press RETURN.
3. If you want channel 1 to light up just press key 1, if channel 3, press key 3 & so on up to 6. If you want it to go OFF just press again.
4. For sensing to be active press 7 for sensing channel 1, 8 for channel 2 & so on.
5. There are two kinds of contact switch: the OPEN or CLOSE type, so choose the corresponding channel as designated by A,B,C,D
6. Starting time for control function: Just type O and then the number of the channel. Then type time. e.g. 6:30 A.M. just type 0630 and the screen will show 06:30. The programme will take care of the : always maintain a 4 digit configuration For 6:30 P.M. just type 1830.
7. For time-off just type F and proceed as before
8. Be sure to set the OUT-IN switch in the OUT position for the real time clock to run. Set it OUT just before you load tape.

PRINTER FUNCTIONS ASSOCIATED WITH CONTROL FUNCTIONS

In some countries the Distributer order from Komtek the Control Function–Basic Printer together as a package. In this case one can have printer function simply by connecting a parallel centronic printer.

OPERATING THE BASIC PRINTER INTERFACE

1) When the printer is already connected to the printer port:
   Turn on the power and the basic instruction of LPRINT" or LLIST will operate the printer.
2) When the printer is not connected to the computer printer port and the computer is turned on or in operation:
   You should then put in the printer connector and then use the following:–

SYSTEM	Press RETURN
* ?	Will appear

Now type the following:- /12367
Type LPRINT" then press return
OM ERROR will appear now proceed the second time & the printer will work.

Notice that every time you press the RETURN key to go to the next line the printer already print on paper giving you no chance of editing. One solution is to use line numbers and the editing can be done using Basic Editing functions. Finally when you are satisfied with it you can use RUN to dump all that you have on the screen onto the printer. To further save you the trouble of writing line numbers you can use AUTO for automatic line number.

OPERATING THE CONTROL FUNCTION

The simplest mode of operationg the control function is to use it to turn on electrical appliances & machines. Suppose the control function channels are connected to 6 control boxes, or even LED's. If you want the first control channel to light up, this is what you do:

1. Turn on the computer.
2. Reset the computer, some version of the same computer with real time clock based on the interrupt (INT) has to be reset by first pressing the switch on the left side of the computer to stop the clock and at the same time press the reset switch on the keyboard for 2 seconds & then release both.
3. Press RETURN key.
4. If you want channel 1 to light up type: POKE 13825, 1
5. Press RETURN
   The LED will immediately light up.
6. Let's say you want channel 3 to light up this is what you type in as a continuation:
   POKE 13825, 4. Press RETURN key. Yes, 4 represent No. 3 channel because all the channels are recalled by a system of binary place value not decimal value.

Channel No.	1	2	3	4	5	6
Binary	00000001	00000010	00000100	00001000	00010000	01000000
Decimal place value	1	2	4	8	16	32

For you information Komtek 1 microcomputer has already 8 control channels. But only 6 is wired for your use externally. If you are an electric engineer or technician you can wire you the 2 extra bonus channels. Another point of interest is that 13825 in the hex system is 3601.

Note: FOR OLDER VERSION OF KOMTEK 1, YOU NEED TO TYPE FIRST POKE 13827, 128 & PRESS RETURN KEY BEFORE YOU TYPE CLAUSE 4.

Occasion arises where one needs to turn on more than one channel at the same time, in this case all you need to do is to superimpose the binary one's & translate into decimal, for example:

If you want to turn on channel 1
Binary 0000 0001
Decimal 1
If you want to turn on channel 3
Binary 0000 0100
Decimal 4
If you want to turn on both channel 1 & 3
0000 0001
0000 0100
0000 0101 which is decimal 5

The principle to do this is, in a more imaginative form, is to imagine all the 0's in the binary train as switches turned off. If you want to turn on any one of them, just put in the number 1 in the required switch position which represent turning on a switch. So the number name of the switch is called by its place value mathematically turning one more than one switch is merely binary addition. So in the decimal system too, it is the aiding of the decimal value!

.You can of course turn on all the switches, in our computer it is the decimal value 63.

Now with all the switches or one of the switch on, how to turn it off?

Type: POKE 13825, 0. & press RETURN key.

For those who purchase Komtek 1 equipped with control function, there is a operating cassette programme to be load into the computer, you do not need to do the "POKING". All you need to do is to press one key for each channel & also to set the clock or alarm & also termination time.

To use the programme, first load the cassette programmes into the computer. After going through the normal loading procedure type RUN, press RETURN key and the switch operating listing will be shown on the screen as a 12 line listing, excluding the real time clock on the top right hand corner.

To set the real time on the top right hand corner you set it from seconds to minutes & then to hour. However this is an interruption of the programme, so you first press BREAK key and proceed as follow:

(Let's suppose the time is now 10:45:32)

a. To set second:
   POKE 16449, 32 press RETURN key.
b. To set minutes:
   POKE 16450, 45 press RETURN key.
c. To set hour:
   POKE 16451, 10 press RETURN key.
d. Type CONTINUE & press RETURN key.
   or Type RUN & press RETURN key.

Now you have the correct time – but, do not turn the computer off otherwise you have to reset again. If you set the time on the computer before you load the control programme you do not need to press the BREAK key; or that you can simply leave the power on so that computer is acting as a clock.

Note that in the above d. when you type CONT you only have part of the listing shown on the screen as your previous setting move the programme on the screen upwards, but the contents is still in the memory, you have not lost anything, just that you are not able to see the complete listing.

However when you type RUN & press RETURN you have a complete listing but you have lost all date that you put in (except the programme). If you are starting a new control function setting it is all right, but when you are in the middle of a programme setting & forget to set the time, then you would have to use the command CONT.

We advise strongly that you should always reset your real time clock first before you will load the control programme to save you the dilemna of choosing CONT or RUN.

Going back to loading the control function after it is loaded type RUN and press RETURN & a 12 line listing will appear on the screen.

SWITCH	OUTPUT	SENSER	OFF IS	TIME ON	TIME OFF
SW 1–	OFF	OFF	OPEN	ØØ:ØØ	ØØ:ØØ
SW 2–	OFF	OFF	OPEN	ØØ:ØØ	ØØ:ØØ
SW 3**	OFF	OFF	OPEN	ØØ:ØØ	ØØ:ØØ
SW 4**	OFF	OFF	OPEN	ØØ:ØØ	ØØ:ØØ
SW 5–	OFF	SW 6–	OFF		
	1	7	A	OF	F
	2	8	B		
	3	9	C		
	4	Ø	D		
	5		6		

The bottom 5 rows of alphabets and numerals represents the control and sensing function switch number as well as the ON-OFF time.

1 represents control function switch 1

2 represents control function switch 2 etc.

7 represents sensing function switch 1

8 represents sensing function switch 2 etc.

NOTE: Control Function switch 5 is represented by 5 and switch 6 is represented by 6, both do not have corresponding sensing functions to match. Of course it holds true for this programme only.

You can also write programme to have switch 5 or 6 to correspond to any of the sensing function channel (or switch). Or alternatively you can use one control channel to correspond to all 4 control functions or vice versa.

To activate the different functions all you need to do is to type the corresponding number or alphabet on the keyboard; typing the same key twice causes it to revert back to the original mode.

One the fourth column is OFF IS. This means that when the senser is ON the way you stop the control function is by choosing the OPEN or CLOSE mode. For example: When switch 1 i.e. SW1 is on & SENSER control is ON and if OFF IS is typed OPEN, control switch 1 will be OFF when you have no sensing lead connected to the sensing channel or that even if it is connected & open circuited it will still be OFF. Thus the control function will operate only when the sensor is CLOSE circuited; if it is OPEN circuited the control function will be OFF.

So when you are boiling water in the morning and if your sensing function is connected to a contact point thermometer it should be of the "OPEN CIRCUIT TYPE" i.e. when certain temperature is reached the circuit will OPEN thus stopping the control function. However if the thermometer you bought is of the "CLOSE-CIRCUIT TYPE" all you need to do is to change the mode by typing A on the keyboard to change the OFF IS for senser 1 to CLOSE.

Every time the sensing function turn OFF the control function it will not revert back by itself again.

This is true for control channel 1& 2. What it means is that in the above example of boiling water, even if the temperature drops after a while when power is turn OFF the control function will not be able to start by itself. Control channel 3 & 4 is special in that it is continuously re-activated by itself every time the sensing function revert back to its original state. This means that when you cook food you can have your food warmed by using that 3rd or 4th control channel & sensing function. Every time when the food gets cold that sensing function will sense the drop via the thermometer and command the control channel to turn on the heat again. This carries on until termination time, as designated by column TIME OFF. So we designate control channel 3 & 4 by two asterisk ** to distinguish its nature from channel 1 & 2. After you have set all the time please make sure that the real time clock is ON. This is done by releasing the OUT-IN switch on the left side of the computer by leaving it in the OUT position. Otherwise the clock will not run!

To set the time of switching on and off for the control function, proceed as follow:

a) Press key 0 and on the screen you will see:
SET SWITCH (1–4)
This means it is asking you what switch (or channel do you want).

b) If you want say channel 2, you simply press key 2 and press RETURN, the screen will show:
TIME:
Now if you want to set the time as 12:30PM, you simply type 12 on the keyboard and the screen will show you
12:
Now type 3 and the screen will show
12:3
Now type Ø the line will disappear!
Because 12:30 is immediately transferred to the TIME ON column.

To set the time OFF press key F and proceed as before. For P.M. add 12 to the time because the computer real time clock runs on HOURS designation basis. e.g. 8 O'clock in the evening is 20:00 HOUR.

By the way, when you have gone through most of the programme and find that there are typing errors, don't press RESET! Just press BACK SPACE to where you made the mistake and re-type again.

Or you can type BREAK and the line in which you write will be erased completely. In case you go way back to the beginning of the programme then you do not want all the intermediate line to be erase; you should then use EDIT functions: please refer to your basic manual for ways of using it.

Finally type RUN and press RETURN. Leave the computer on until the control function has done its job.

You can write programme to convert the real time clock on the top right hand corner of the screen to drive big time display on the screen. So that for those of you who are short sighted or that you TV is every far away you can still see the time from afar.

Priorities:

Channel 1 to channel 4 can all be operated manually and in conjunction with sensing function.
In the case of channel 1 & 2. The sensor only does its duty for once and stops.
In the case of channel 3 & 4 the sensor works on the control function ON & OFF for unlimited number of times.
However once you set the time of ON or OFF the proper procedure is to set control and sensing channel both in the OFF mode except the sensing mode indicated by the column OFF IS.
When the time comes the machine will automatically turns both control and sensing function ON for duty till it is turned off by the programmed time.
(However before the set time you can still mainpulate channel 3 & 4 by hand after the set time you can also mainpulated manually as an override)

```
10 CLS:GOTO570
20 CLS:CT=0:PRINT:POKE 13827,137
30 T$="17A28B39C40D":TI=PEEK(16451)*16+PEEK(16450)
40 POKE 13825,0
50 PRINT"TO SEEK INSTRUCTION:- PRESS I"
60 PRINT:PRINT"                                          TIME      TIME
SWITCH  OUTPUT  SENSER  OFF IS    ON       OFF
SW 1 --  OFF     OFF     OPEN    00:00    00:00
SW 2 --  OFF     OFF     OPEN    00:00    00:00
SW 3 **  OFF     OFF     OPEN    00:00    00:00
70 PRINT"SW 4 **  OFF     OFF     OPEN    00:00    00:00
SW 5 --  OFF     SW 6 --  OFF
          1        7        A        0        F
          2        8        B                 -
          3        9        C
          4        0        D
          5                 6"
80 DIM SW(6,5):FOR A=1 TO 4:SW(A,3)=1:NEXT
90 GOSUB 100:GOTO 200
100 GOSUB 520:TC=PEEK(16451)*16+PEEK(16450):IF INT(TC-TI)>0 THEN GOSUB 350
110 IF SW(1,2)+SW(2,2)=0 THEN 150 ELSE FOR A=1 TO 2:IF SW(A,2)=0 OR SW(A,1)=0 TH
EN 140
120 B=PEEK(13826) AND 2[(A+3):IF (B>0)*-1<>SW(A,3) THEN 140
130 X=A:GOSUB 330
140 NEXT
150 IF SW(3,2)+SW(4,2)=0 THEN RETURN ELSE FOR A=3 TO 4
160 IF SW(A,2)=0 THEN 190
170 IF ((PEEK(13826) AND 2[(A+3))>0)*-1=SW(A,3) THEN IF SW(A,1)=1 THEN X=A:GOSUB
 330:GOTO 190
180 IF ((PEEK(13826) AND 2[(A+3))>0)*-1<>SW(A,3) THEN IF SW(A,1)=0 THEN GOSUB 54
0:SW(A,1)=1:CT=CT+2[(A-1):PRINT@265+A*64,"ON ";:POKE 13825,CT:GOSUB 550
190 NEXT:RETURN
200 K$=INKEY$:IF K$="" THEN 90
210 IF K$="0" OR K$="F" THEN 420
220 IFK$="I"THEN570
230 IF K$<>"5" AND K$<>"6" THEN 260 ELSE S=6:IF K$="5" THEN S=5
240 IF SW(S,1)=0 THEN CT=CT+2[(S-1):P$="ON ":SW(S,1)=1 ELSE CT=CT-2[(S-1):P$="OF
F":SW(S,1)=0
250 POKE 13825,CT:PRINT @500+S*17,P$;
260 FOR A=1 TO 12:IF MID$(T$,A,1)<>K$ THEN 320
270 C=INT((A-1)/3)+1:B=A-(C-1)*3
280 IF SW(C,B)=1 THEN SW(C,B)=0:P$="CLOSE" ELSE SW(C,B)=1:P$="OPEN "
290 IF B<3 THEN IF SW(C,B)=1 THEN P$="ON " ELSE P$="OFF"
300 IFB=1 THEN IF SW(C,B)=1 THEN CT=CT+2[(C-1):POKE 13825,CT ELSE X=C:GOSUB 330
310 PRINT@257+C*64+B*8,P$;:GOTO 90
320 NEXT:GOTO 90
```

```
330 GOSUB 540:CT=CT-2[(X-1):POKE 13825,CT:PRINT@265+X*64,"OFF";
340 GOSUB 550:SW(X,1)=0:RETURN
350 GOSUB 540:FOR A=1 TO 4:TI=TC:IF SW(A,4)=SW(A,5) THEN 410
360 IF SW(A,1)=1 THEN 390
370 IF SW(A,4)<>TI THEN 410
380 SW(A,1)=1:PRINT@265+A*64,"ON ";:CT=CT+2[(A-1):POKE 13825,CT:GOTO 410
390 IF SW(A,5)<>TI THEN 410
400 SW(A,1)=0:PRINT@265+A*64,"OFF";:CT=CT-2[(A-1):POKE 13825,CT
410 NEXT:GOTO 550
420 S$=K$:PRINT@962,"SET SWITCH  (1-4) ";
430 GOSUB 100:K$=INKEY$:IF K$>"4" OR K$<"1" THEN 430 ELSE S=VAL(K$)
440 PRINT@962,CHR$(30);"TIME :  ";:ST$="":TS=0:ST=0
450 GOSUB 100:K$=INKEY$:IF K$>"9" OR K$<"0" THEN 450
460 PRINT K$;
470 ST$=ST$+K$:IFLEN(ST$)=2 THEN TS=TS+1:ST=ST*16+VAL(ST$):PT$(TS)=ST$:ST$="";IF
 TS<2 THEN PRINT ":";:GOTO 450 ELSE IF VAL(PT$(1))>23 ORVAL(PT$(2))>59 THEN 440
ELSE 490



480 GOTO 450
490 IF S$="0" THEN SW(S,4)=ST:SL=1 ELSE SW(S,5)=ST:SL=2
500 PRINT CHR$(29);CHR$(30);
510 PRINT @280+S*64+SL*9,PT$(1);":";PT$(2);:GOTO 90
520 GOSUB 540:GOSUB 540:PRINT @181,"";:Z=PEEK(16451):GOSUB 560:PRINT ":";:Z=PEEK
(16450):GOSUB 560:PRINT ":";:Z=PEEK(16449):GOSUB 560:GOTO 550
530 GOSUB 550:RETURN
540 C1=PEEK(16416):C2=PEEK(16417):RETURN
550 POKE 16416,C1:POKE 16417,C2:RETURN
560 PRINT RIGHT$("0"+RIGHT$(STR$(Z),LEN(STR$(Z))-1),2);:RETURN
570 PRINT" INSTRUCTION FOR CONTROL & SENSING FUNCTION-TIMER PROGRAMME
580 PRINT" THIS IS A DEMONSTRATION PROGRAMME FOR YOU TO FAMILIARIZE        YOURSE
LF WITH OUR CONTROL SYSTEM OF KOMTEK I. YOU CAN TURN ON 6 APPLIANCES BY 3 WAYS
590 PRINT:PRINT
600 PRINT"1ST  SIMPLE SWITCH ACCORDING TO THE # ON KEYBOARD
        I.E. : SW 5, SW 6
610 PRINT"2ND  CONNECTED WITH SENSING FUNCTION STATUS                      3RD  C
ONTROLED BY PRESET TIMER (SW 1 TO SW 4)
620 PRINT:PRINT:PRINT"THE SITUATION OF THE SWITCH, TIMER, & SENSOR ARE SHOWN IN
THE   TABLE BELOW:
630 FORT=0TO2000:NEXT
640 CLEAR255
650 GOTO20
```

DISPLAYING TIME

If you do not intent to load or use any control functions but just to use the computer as a clock the following are the procedures:

a) Look at your watch and set the correct time by POKE method suppose the time is 14:25:42 i.e. 2 o'clock in the afternoon Twenty Five Minutes and Fourty-two seconds.

b) To set seconds:
POKE 16449, 42
Press RETURN key
To set Minutes
POKE 16450, 25
press RETURN key
The clock in the computer is already running and that gives you any amount of time to type in the display command.

c) To tell the computer to display the time on the screen type in the following programme:

```
10 PRINT @56, " ";: FOR A = 16451 to 16449 STEP-1
20 AS = STR$ (PEEK (A) ) : PRINT RIGHT $
  ("Ø"+RIGHT $ (AS, LEN (AS)-1), 2);
30 IF A () 16449 THEN PRINT ":";
40 NEXT
50 GOTO 1Ø
```

To execute the above programme, type RUN and press RETURN key. The time will then be displayed on the screen.

d) For your own convenience we suggest that you load the programme on tape so that you do not have to go through the tedious procedure of typing out the programme.

In case you wish to display the time on one more line below, just add 64 to 56 i.e. 120. In line 10 PRINT @56 now becomes PRINT @120. Following the same reasoning if you like to display 5 spaces further to the left then type PRINT @115.

CONTROL FUNCTIONS

FURTHER EXPANSION

Komtek 1 fitted with control functions may be expanded to a total of 24 channels without any additional hardware internally. One only need to use a 34 PIN flat cable & connector to tap the control channels from the printer port. This is because the printer port is only wired for 12 channels, plus the 6 existing control function channels and the existing 4 sensing channel makes a 22 channel control & sensing function machine. So for those who do not wish to do modifications he can still have a ready 22 channels control functions.

To get the last two channels one only needs to connect two wires to the printer port:

1. Connect printer port pin NO. 19 to pin NO. 14 of control chip 8255 by soldering a wire and also cut the 5V power leading to pin No. 14 of 8255.
2. Connect printer port pin NO. 27 to pin NO. 24 of 8255.

The above connection applies to computer which has a mainboard designated from KT1001-01-62 up to KT1001-01-64CI. For mainboard which are of the KT1001-01-64D version and later version all one need is to apply a drop of solder to connect two half moon shaped pieces to connect up the wire laid into the board and also to cut the power to pin NO. 14 of 8255 marked by a cross X over the wire, with a sharp knife.

For those who buy control functions of 64C version and later a simple demonstration programme for the 20 control function outlet is supplied with the package. This simple programme is stored after the main demonstration programme on the demonstration tape. One can list out the programme to gain an insight of how it is written. In this programme there are no sensing functions. Of course you can add them into the programme yourself.

One can also change some of the channels into sensing channels by adding a series resistor to output of 2.7 to 4.7K, and a boot up resistor of 270 to 100 OHMS. The boot up resistor is connected to the 5V + power supply.

- CAUTION: One must be well versed with electronic circuits to do this conversion. The manufacturer & Distributor will not entertain any claim arising from damages due to control function expansion modifications. The user is advised to use the 22 channels available without any soldering.
- IMPORTANT: All control function interaction with the outside world where higher voltage are involved must be properly insolated by optical couplers or relays, or thryisters. Any leakage may caused the computer to burn out! In order not to unduly overload the power supply when all 24 control channels are. Operating at the same time all the time, we strongly advise users to feed the signal through a transistor so that the load is on the transistor & that there are practically very little current drain on the 8255 chip. One separate transistor for each channel should be used.

FULL EXPANSION

For those who are using Komtek for more process & industrial & building controls the expansion circuit is recommended. One can use expander boards with banks of 8255 and stack up to a total of 255 channels. A separate power supply is needed. However the supply of these systems are on a contract & custom made basis. In many countries these 255 channel control system Komtek are marked under a different distributor than those who market personal or home computer. Also it may appear on the market under a different brand name i.e. on a OEM basis.

SIGNAL CONVERSION

Sometimes it may be helpful to have a A/D i.e. analogue to Digital converter to convert your signal into quantitative Digital signal for the computer to analyse. For example if you have a certain voltage to monitor you can set a threshold digitally so that if the voltage exceed or drop below a certain voltage the control function would operate according to a certain programme mode. Komtek's simple A/D converter plugs into the expander port. If you have a printer interface hooked up onto the expander port you can get an extra connection by putting a second female parallel output connector or use our multi expander output cable.

Expander pin Number	8355 pin Number	Programming Number	Demonstration Programme Designation
1	25	POKE 13825, 128	428
3	4	POKE 13824, 1	1
5	3	POKE 13824, 2	2
7	2	POKE 13824, 4	4
9	1	POKE 13824, 8	8
11	40	POKE 13824, 16	16
13	39	POKE 13824, 32	32
15	38	POKE 13824, 64	64
17	37	POKE 13824, 128	128
19	14	POKE 13826, 1	601
21	17	POKE 13826, 8	608
23	16	POKE 13826, 4	604
25	15	POKE 13826, 2	602
27	24	POKE 13825, 64	364
29			
31			
33			
Existing Control Outlets Fitted On The Computer	Control channel 1	POKE 13825, 01	301
	Control channel 2	POKE 13825, 02	302
	Control channel 3	POKE 13825, 04	304
	Control channel 4	POKE 13825, 08	308
	Control channel 5	POKE 13825, 16	316
	Control channel 6	POKE 13825, 32	332
Existing Sensing Outlets Fitted On The Computer	Sensing Channel 1	POKE 13826, 16	616
	Sensing Channel 2	POKE 13826, 32	632
	Sensing Channel 3	POKE 13826, 64	664
	Sensing Channel 4	POKE 13826, 128	728

CONNECTING THE COMPUTER

The following diagramme shows the connection points of Komtek 1 and is self explanatory.



```
1 FORT=0TO16:PRINT:NEXT
2 PRINT"THIS IS KOMTEK CONTROL PROGRAMME. YOU  CAN USE IT TO CONTROL  20 DIFFERE
NT APPLIANCES. "
3 PRINT:PRINT:
5 PRINT"THE APPLIANCES ARE NUMBERED AS FOLLOWS:"
6 PRINT" 1        2        4        8       16       32     0 64      128"
7 PRINT"301      302      304      308      316      332      364 (  428"
8 PRINT"601      602      604      608
9 PRINT:PRINT"TO TURN THEM OFF, TYPE 999":PRINT"TO RECALL TABLE,TYPE  1000":PRIN
T
10 POKE13827,128
4 PRINT"WHICH LAMP (OR APPLIANCES) DO YOU WANT":
20 INPUTA
21 IFA>=300ANDA <356THEN100
22 IFA>=600AND A<856THEN200
23 IFA=999THEN1000:
24 IFA=1000THEN2E
25 IFA>=0ANDA<256THEN30
26 PRINT"ILLOGICAL INPUT":GOTO20
30 POKE13824,A
40 GOTO500
100 B=A-300
110 POKE13825,B
130 GOTO500
200 C=A-600
210 POKE13826,C
500 PRINT:PRINT"WHAT OTHER APPLIANCES WOULD YOU LIKE TO TURN ON?":PRINT:
550 GOTO20
1000 PRINT"TURN OFF THE 1st ROW OF LAMPS BY INPUT 0"
1010 PRINT"TURN OFF THE 2nd ROW BY INPUT 300"
1020 PRINT"TURN OFF THE 3rd ROW BY INPUT 600"
1040 GOTO20
```